Jadavpur University
Department of Physics

Numerical Solutions using the WKB Approximation

The Quantization Condition

$$\int_{x_L}^{x_R} \sqrt{2m(E - V(x))}\, dx = (n + \alpha)\, \pi\hbar$$

Submitted by:
Mohak Das

Supervisor:
Dr. Asim Kumar Ghosh

Class: PG-I
Class Roll No: 002520702028
Exam Roll No: PPHD0261034

Computational Physics Lab

Abstract

This report presents the numerical application of the semiclassical Wentzel-Kramers-Brillouin (WKB) approximation to solve the time-independent Schrödinger equation for various one-dimensional potentials. A generalized Fortran 90 solver was developed using the Bohr-Sommerfeld quantization condition and the Bisection root-finding method. The solver was tested against eleven distinct potentials, including harmonic, infinite well, and anharmonic systems. The numerical results show excellent agreement with exact analytical values for the Harmonic Oscillator ($> 99.99\%$ accuracy) and the Infinite Square Well, validating the solver's robustness. For anharmonic potentials where exact solutions are not available, the results exhibit the expected physical scaling behavior.

Contents

1 Introduction

1.1 Main Idea

The Wentzel–Kramers–Brillouin (WKB) approximation is a semiclassical method used in quantum mechanics to solve the time-independent Schrödinger equation. It is particularly useful for systems where the potential energy $V(x)$ varies slowly compared to the de Broglie wavelength of the particle. While exact analytical solutions are rare for arbitrary potentials, the WKB method provides a powerful tool to approximate the wavefunction and, crucially, to calculate accurate energy eigenvalues for bound states without solving the full differential equation.

1.2 The Approximation

The method starts by assuming the wavefunction $\psi(x)$ can be written in the form of a plane wave with a slowly varying amplitude and phase. The ansatz is:

$$\boxed{\psi(x) \approx \frac{C}{\sqrt{p(x)}} e^{\pm \frac{i}{\hbar} \int p(x)\, dx}}$$

where $p(x) = \sqrt{2m(E - V(x))}$ is the classical momentum of the particle. This approximation holds true in regions where the potential $V(x)$ changes slowly enough that the change in the particle's momentum over one wavelength is small.

1.3 The Quantization Condition

For bound states, the particle is confined between turning points (where $E = V(x)$). By matching the oscillatory solutions inside the potential well with the exponentially decaying solutions in the classically forbidden regions (using Airy functions near the turning points), we derive a constraint on the action integral. The general Bohr-Sommerfeld quantization condition is:

$$\boxed{\int_{x_1}^{x_2} p(x)\, dx = \int_{x_1}^{x_2} \sqrt{2m(E - V(x))}\, dx = (n + \alpha)\, \pi\hbar}$$

where x_1 and x_2 are the boundaries of the motion, n is an integer quantum number, and α is a phase constant determined by the boundary conditions (e.g., rigid walls vs. soft potential slopes).



Figure 1: Schematic of the WKB approximation. The particle is confined between classical turning points x_L and x_R, where the energy E intersects the potential $V(x)$.

2 Numerical Implementation

2.1 Global Constants and Precision

To ensure numerical stability and physical consistency across all problems, a shared module `constants_mod` is used. This module defines the double-precision floating-point kind (`real64`) and sets the atomic units ($\hbar = 1, m = 1$).

```
module constants_mod
    implicit none
    integer, parameter :: dp = selected_real_kind(15, 307)
    real(dp), parameter :: h_bar = 1.0_dp
    real(dp), parameter :: mass = 1.0_dp
    real(dp), parameter :: pi = 3.14159265358979323846264338327 95_dp
contains
end module constants_mod
```

Listing 1: Global Constants Module

2.2 The General WKB Solver

The core logic is encapsulated in the `wkb_toolbox` module.

The WKB Solver

The computational approach treats the quantization condition as a root-finding problem. For a given quantum number n and phase factor α, we seek the energy E that satisfies:

$$f(E) = \int_{x_L}^{x_R} \sqrt{2m(E - V(x))}\, dx - (n + \alpha)\pi\hbar = 0$$

The solver employs the **Bisection Method**, a robust root-finding algorithm. It iteratively narrows the energy bracket $[E_{min}, E_{max}]$ by evaluating the action integral at the midpoint and checking the sign of the residual. This method ensures convergence to the global solution for the monotonic action function.

Choice of Integration Scheme: Trapezoidal Rule

The action integral is computed using the **Composite Trapezoidal Rule**. This method was selected specifically for its stability in the context of WKB integrals:

- **Boundary Behavior:** The integrand $p(x) = \sqrt{2m(E - V(x))}$ vanishes identically at the classical turning points (x_L, x_R). The Trapezoidal rule incorporates these zero-valued endpoints naturally without requiring special handling.
- **Singularity Avoidance:** The derivative of the momentum function becomes singular at the turning points (since $\frac{d}{dx}\sqrt{x} \to \infty$ as $x \to 0$). Higher-order methods like Simpson's rule or Gaussian quadrature can be sensitive to such non-smooth behavior at the boundaries. The Trapezoidal rule, which relies solely on function evaluations rather than derivatives, remains numerically stable and sufficiently accurate for this class of integrals.

```fortran
module wkb_toolbox
    use constants_mod
    implicit none

    abstract interface
        subroutine turn_point_func(E, x_L, x_R)
            use constants_mod
            real(dp), intent(in)  :: E
            real(dp), intent(out) :: x_L, x_R
        end subroutine turn_point_func
    end interface
contains

    ! ------------------------------------------------------------------
    ! Generic Integrator: Calculates Integral p(x) dx
    ! ------------------------------------------------------------------
    function wkb_integral(E, V, x_in_L, x_in_R) result(action)
        real(dp), external :: V
        real(dp), intent(in) :: E
        real(dp), intent(in) :: x_in_L, x_in_R
        real(dp) :: action

        real(dp) :: x_left, x_right, x, dx, p_x
        integer :: i
        integer, parameter :: N_STEPS_DEFAULT = 2000

        x_left = x_in_L
        x_right = x_in_R

        ! Integrate (Trapezoidal Rule to preserve accuracy at the turning points)
        dx = (x_right - x_left) / real(N_STEPS_DEFAULT, dp)
        action = 0.0_dp

        x = x_left
        if (E > V(x)) then
            p_x = sqrt(2.0_dp * mass * (E - V(x)))
            action = action + 0.5_dp * p_x
        end if

        do i = 1, N_STEPS_DEFAULT - 1
            x = x_left + i * dx
            if (E > V(x)) then
                p_x = sqrt(2.0_dp * mass * (E - V(x)))
                action = action + p_x
            end if
        end do

        x = x_right
        if (E > V(x)) then
            p_x = sqrt(2.0_dp * mass * (E - V(x)))
            action = action + 0.5_dp * p_x
        end if

        action = action * dx
    end function wkb_integral
```

```fortran


    ! -------------------------------------------------------------------
    ! Generic Root Finder: Solves for Energy
    ! -------------------------------------------------------------------
    function find_eigenenergy(n, V, alpha, tp_formula, E_max) result(
   E_root)
        real(dp), external :: V
        integer, intent(in) :: n
        real(dp), intent(in) :: alpha
        real(dp), intent(in), optional :: E_max
        procedure(turn_point_func) :: tp_formula
        real(dp) :: E_root

        real(dp) :: E_min, E_max_val, E_mid
        real(dp) :: target_action, current_action
        real(dp) :: calc_L, calc_R
        integer :: iter
        real(dp), parameter :: TOL_E = 1.0e-7_dp

        ! WKB quantization condition:
        ! \int_{x_1}^{x_2} \sqrt{2m(E - V(x))} dx = (n + \alpha) \pi \hbar
        target_action = (real(n, dp) + alpha) * pi * h_bar

        E_min = 0.0001_dp

        if (present(E_max)) then
            E_max_val = E_max
        else
            E_max_val = 500.0_dp
        end if

        ! Bisection Algorithm
        do iter = 1, 1000
            E_mid = 0.5_dp * (E_min + E_max_val)

            call tp_formula(E_mid, calc_L, calc_R)
            current_action = wkb_integral(E_mid, V, x_in_L=calc_L, x_in_R=
   calc_R)

            if (abs(current_action - target_action) < TOL_E) exit

            if (current_action < target_action) then
                E_min = E_mid ! Energy too low
            else
                E_max_val = E_mid ! Energy too high
            end if
        end do

        E_root = E_mid
    end function find_eigenenergy

end module wkb_toolbox
```

Listing 2: WKB Toolbox Implementation

3 Solution to the Assignment Problems

3.1 Problem 1: One-dimensional Half-Harmonic Potential

- **Potential:** $V(x) = \frac{1}{2}m\omega^2 x^2$ for $x \geq 0$, ∞ otherwise.
- **Quantization Condition Used:** One Rigid Wall, One Soft Turning Point.

$$\int_0^{x_R} \sqrt{2m(E - V(x))}\, dx = \left(n - \frac{1}{4}\right)\pi\hbar \quad (n = 1, 2, \cdots)$$

- **Reason:** The potential has a "hard wall" at $x = 0$ (infinite potential), forcing the wavefunction to vanish ($\psi(0) = 0$). The other boundary x_R is a "soft" turning point where the potential slopes gently. The phase shift contribution is 0 from the hard wall and $-\pi/4$ from the soft turning point, leading to the $(n - 1/4)$ factor.
- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies E = \frac{1}{2}m\omega^2 x^2$$
$$\implies x^2 = \frac{2E}{m\omega^2}$$
$$\implies x = \pm\sqrt{\frac{2E}{m\omega^2}}$$

using $m = 1$ and $\omega = 1$, we get the right turning point $\boxed{x_R = \sqrt{2E}}$. Since this is a Half-Harmonic potential, the left turning point $\boxed{x_L = 0}$.

- **Code:**

```fortran
!=============================================================
! File     : prob01_half_harmonic.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Half Harmonic Oscillator
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!=============================================================


module prob01_physics
    use constants_mod
    implicit none
    real(dp), parameter :: omega = 1.0_dp
    real(dp), parameter :: alpha = -0.25_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
```

```fortran
        real(dp) :: res
        res = 0.5_dp * mass * (omega**2) * x**2
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = 0.0_dp
        x_R = sqrt(2.0_dp * E)
    end subroutine turning_points

end module prob01_physics


program prob01_driver
    use wkb_toolbox
    use prob01_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "------------------------------------------------"
    write(*, '(A)') "Problem 1: Half Harmonic Oscillator"
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 1, 5
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob01_driver
```

- **Output:**

```
------------------------------------------------
Problem 1: Half Harmonic Oscillator
------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
------------------------------------------------
The five lowest eigen energies:

n =  1    E = 1.500006232
```

```
n =  2    E = 3.500014658
n =  3    E = 5.500022967
n =  4    E = 7.500031394
n =  5    E = 9.500039820
```

3.2 Problem 2: One-dimensional Harmonic Potential

- **Potential:** $V(x) = \frac{1}{2}m\omega^2 x^2$.
- **Quantization Condition Used:** Two Soft Turning Points.

$$\int_{x_L}^{x_R} \sqrt{2m(E - V(x))}\, dx = \left(n + \frac{1}{2}\right)\pi\hbar \quad (n = 0, 1, \cdots)$$

- **Reason:** The particle is confined by the potential on both sides without any hard walls. Both turning points x_L and x_R are "soft" (linear turning points), each contributing a phase shift of $-\pi/4$. The total phase loss is $\pi/2$, resulting in the standard $(n + 1/2)$ term.
- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies E = \frac{1}{2}m\omega^2 x^2$$
$$\implies x^2 = \frac{2E}{m\omega^2}$$
$$\implies x = \pm\sqrt{\frac{2E}{m\omega^2}}$$

using $m = 1$ and $\omega = 1$, we get the right turning point $\boxed{x_R = \sqrt{2E}}$ and the left turning point $\boxed{x_L = -\sqrt{2E}}$.

- **Code:**

```fortran
!============================================================
! File     : prob02_1D_harmonic.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Harmonic Oscillator
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================


module prob02_physics
    use constants_mod
    implicit none
    real(dp), parameter :: omega = 1.0_dp
    real(dp), parameter :: alpha = 0.5_dp
```

```fortran

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = 0.5_dp * mass * (omega**2) * x**2
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = - sqrt(2.0_dp * E)
        x_R = sqrt(2.0_dp * E)
    end subroutine turning_points

end module prob02_physics


program prob02_driver
    use wkb_toolbox
    use prob02_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "----------------------------------------------"
    write(*, '(A)') "Problem 2: Harmonic Oscillator"
    write(*, '(A)') "----------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 0, 4
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob02_driver
```

- **Output:**

```
----------------------------------------------
Problem 2: Harmonic Oscillator
----------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
------------------------------------------------
The five lowest eigen energies:
```

```
n =   0     E =  0.5000059184
n =   1     E =  1.500017757
n =   2     E =  2.500029595
n =   3     E =  3.500041433
n =   4     E =  4.500053271
```

3.3 Problem 3: One-dimensional Infinite Square Well Potential

- **Potential:** $V(x) = 0$ for $-L \leq x \leq L$, ∞ otherwise.
- **Quantization Condition Used:** Two Rigid Walls.

$$\int_{-L}^{L} \sqrt{2mE}\, dx = n\pi\hbar \quad (n = 1, 2, \cdots)$$

- **Reason:** The particle is confined strictly between $x = -L$ and $x = L$ by infinite potential barriers. The wavefunction must be exactly zero at both boundaries. There are no "soft" turning points region to generate the $\pi/4$ phase shifts, so the phase factor α is zero.
- **Code:**

```
!============================================================
! File     : prob03_infinite_square_well.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Infinite Square Well
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================


module prob03_physics
    use constants_mod
    implicit none
    real(dp), parameter :: width = 1.0_dp
    real(dp), parameter :: alpha = 0.0_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = 0.0_dp
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = -width
        x_R = width
```

```fortran
    end subroutine turning_points

end module prob03_physics


program prob03_driver
    use wkb_toolbox
    use prob03_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "----------------------------------------------------"
    write(*, '(A)') "Problem 3: Infinite Square Well"
    write(*, '(A)') "----------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "----------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 1, 5
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob03_driver
```

- **Output:**

```
----------------------------------------------------
Problem 3: Infinite Square Well
----------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
----------------------------------------------------
The five lowest eigen energies:

n =  1    E = 1.233700532
n =  2    E = 4.934802246
n =  3    E = 11.10330502
n =  4    E = 19.73920887
n =  5    E = 30.84251355
```

3.4 Problem 4: Infinite Potential Well with Slanted Base

- **Potential:** $V(x) = \frac{V_0}{L}x$ for $0 \leq x \leq L$, ∞ otherwise.

- **Quantization Condition Used:** One Rigid Wall, One Soft Turning Point.

$$\int_0^{x_R} \sqrt{2m(E - V(x))}\, dx = \left(n - \frac{1}{4}\right)\pi\hbar \quad (n = 1, 2, \cdots)$$

- **Reason:** For the low-lying energy states calculated ($E < V_0$), the particle turns back at a point $x_R < L$ where $E = V(x)$. Thus, the system effectively has one hard wall at $x = 0$ and one soft classical turning point at x_R.

- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = \frac{V_0}{L}x$$

$$\implies \quad x = \frac{L}{V_0}E$$

Thus, the right turning point $\boxed{x_R = \frac{L}{V_0}E}$.

- **Code:**

```fortran
!============================================================
! File     : prob04_infinite_potential_well_slanted_base.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Infinite Square Well With Slanted Base
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================



module prob04_physics
    use constants_mod
    implicit none
    real(dp), parameter :: width = 1.0_dp
    real(dp), parameter :: V0 = 20000_dp
    real(dp), parameter :: E_max = V0
    real(dp), parameter :: alpha = -0.25_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = (V0/width) * x
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R
```

```fortran

        x_L = 0.0_dp
        x_R = (width/V0) * E
    end subroutine turning_points

end module prob04_physics


program prob04_driver
    use wkb_toolbox
    use prob04_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "-------------------------------------------------------"
    write(*, '(A)') "Problem 4: Infinite Potential Well With Slanted Base"
    write(*, '(A)') "-------------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "-------------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "


    do n = 1, 5
        energy = find_eigenenergy(n, potential, alpha, turning_points, E_max)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob04_driver
```

- **Output:**

```
-------------------------------------------------------
Problem 4: Infinite Potential Well With Slanted Base
-------------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
-------------------------------------------------------
The five lowest eigen energies:

n =  1    E = 1356.894020
n =  2    E = 2387.062474
n =  3    E = 3226.464475
n =  4    E = 3967.582217
n =  5    E = 4644.805121
```

3.5 Problem 5: Infinite Potential Well with Parabolic Base

- **Potential:** $V(x) = \frac{V_0}{L^2}x^2$ for $0 \le x \le L$, ∞ otherwise.
- **Quantization Condition Used:** One Rigid Wall, One Soft Turning Point.

$$\int_0^{x_R} \sqrt{2m(E - V(x))}\, dx = \left(n - \frac{1}{4}\right)\pi\hbar \quad (n = 1, 2, \cdots)$$

- **Reason:** Similar to the slanted well, for energies $E < V_0$, the particle is confined by the hard wall at $x = 0$ and a soft turning point x_R formed by the parabolic potential.
- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = \frac{V_0}{L^2}x^2$$

$$\implies \quad x^2 = \frac{L^2}{V_0}E \implies \quad x = \pm\sqrt{\frac{L^2}{V_0}E}$$

Thus, the right turning point $\boxed{x_R = L\sqrt{\frac{E}{V_0}}}$.

- **Code:**

```fortran
!============================================================
! File     : prob05_parabolic_base.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Infinite Potential Well With
   Parabolic Base
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================


module prob05_physics
    use constants_mod
    implicit none
    real(dp), parameter :: width = 1.0_dp
    real(dp), parameter :: V0 = 20000_dp
    real(dp), parameter :: E_max = V0
    real(dp), parameter :: alpha = -0.25_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = (V0/(width**2)) * x**2
    end function potential

```

```fortran
    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = 0.0_dp
        x_R = sqrt(E/V0) * width
    end subroutine turning_points

end module prob05_physics


program prob05_driver
    use wkb_toolbox
    use prob05_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "-------------------------------------------------------"
    write(*, '(A)') "Problem 5: Infinite Potential Well With Parabolic Base"
    write(*, '(A)') "-------------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "-------------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 1, 5
        energy = find_eigenenergy(n, potential, alpha, turning_points, E_max)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob05_driver
```

- **Output:**

```
-------------------------------------------------------
Problem 5: Infinite Potential Well With Parabolic Base
-------------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
-------------------------------------------------------
The five lowest eigen energies:

n =  1    E = 300.0012653
n =  2    E = 700.0029307
n =  3    E = 1100.004596
n =  4    E = 1500.006280
```

```
n =   5     E =  1900.007964
```

3.6 Problem 6: $V(x) = |x|$

- **Potential:** $V(x) = |x|$.
- **Quantization Condition Used:** Two Soft Turning Points.

$$\int_{x_L}^{x_R} \sqrt{2m(E - |x|)}\, dx = \left(n + \frac{1}{2}\right)\pi\hbar \quad (n = 0, 1, \cdots)$$

- **Reason:** The potential is continuous and confining on both sides ($x \to \pm\infty$). There are no infinite walls, so the standard WKB rule for two soft turning points applies.
- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = |x|$$

$E = x$ for $x \geq 0$ and $E = -x$ for $x < 0$. Thus, the right turning point $\boxed{x_R = E}$ and the left turning point $\boxed{x_L = -E}$.

- **Code:**

```fortran
!============================================================
! File     : prob06_modx.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for |x| potential
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================


module prob06_physics
    use constants_mod
    implicit none
    real(dp), parameter :: alpha = 0.5_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = abs(x)
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = -E
```

```fortran
        x_R = E
    end subroutine turning_points

end module prob06_physics


program prob06_driver

    use wkb_toolbox
    use prob06_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "------------------------------------------"
    write(*, '(A)') "Problem 6: |x| potential"
    write(*, '(A)') "------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 0, 4
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob06_driver
```

- **Output:**

```
------------------------------------------
Problem 6: |x| potential
------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
------------------------------------------
The five lowest eigen energies:

n =  0    E = 0.8853471922
n =  1    E = 1.841596308
n =  2    E = 2.588770794
n =  3    E = 3.239755918
n =  4    E = 3.830674676
```

3.7 Problem 7: $V(x) = |x^3|$

- **Potential:** $V(x) = |x^3|$.

- **Quantization Condition Used:** Two Soft Turning Points.

$$\int_{x_L}^{x_R} \sqrt{2m(E - |x^3|)}\, dx = \left(n + \frac{1}{2}\right)\pi\hbar \quad (n = 0, 1, \cdots)$$

- **Reason:** The potential is continuous and confining on both sides ($x \to \pm\infty$). There are no infinite walls, so the standard WKB rule for two soft turning points applies.

- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = |x^3|$$

$E = x^3$ for $x \geq 0$ and $E = -x^3$ for $x < 0$. Thus, the right turning point $\boxed{x_R = \sqrt[3]{E}}$ and the left turning point $\boxed{x_L = -\sqrt[3]{E}}$.

- **Code:**

```
!============================================================
! File     : prob07_modx_cube.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for |x^3| potential
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================


module prob07_physics
    use constants_mod
    implicit none
    real(dp), parameter :: alpha = 0.5_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = abs(x**3)
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = - E ** (1.0_dp/3.0_dp)
        x_R = E ** (1.0_dp/3.0_dp)
    end subroutine turning_points

end module prob07_physics


program prob07_driver
```

```
    use wkb_toolbox
    use prob07_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "--------------------------------------------"
    write(*, '(A)') "Problem 7: |x^3| potential"
    write(*, '(A)') "--------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar
    = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "--------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)
   ') " "

    do n = 0, 4
        energy = find_eigenenergy(n, potential, alpha, turning_points
   )
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob07_driver
```

- **Output:**

```
-----------------------------------------------
Problem 7: |x^3| potential
-----------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
-----------------------------------------------
The five lowest eigen energies:

n =  0    E = 0.6075053185
n =  1    E = 2.270364679
n =  2    E = 4.190965764
n =  3    E = 6.275780640
n =  4    E = 8.484790383
```

3.8 Problem 8: $V(x) = x^4$

- **Potential:** $V(x) = x^4$.
- **Quantization Condition Used:** Two Soft Turning Points.

$$\int_{x_L}^{x_R} \sqrt{2m(E - x^4)}\, dx = \left(n + \frac{1}{2}\right) \pi\hbar \quad (n = 0, 1, \cdots)$$

- **Reason:** The potential is continuous and confining on both sides ($x \to \pm\infty$). There are no infinite walls, so the standard WKB rule for two soft turning points applies.

- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = x^4$$

Thus, the right turning point $\boxed{x_R = \sqrt[4]{E}}$ and the left turning point $\boxed{x_L = -\sqrt[4]{E}}$.

- **Code:**

```
!==============================================================
! File     : prob08_quartic_potential.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Quartic (x^4) potential
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!==============================================================


module prob08_physics
    use constants_mod
    implicit none
    real(dp), parameter :: alpha = 0.5_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = x**4
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = - sqrt(sqrt(E))
        x_R = sqrt(sqrt(E))
    end subroutine turning_points

end module prob08_physics


program prob08_driver
    use wkb_toolbox
    use prob08_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "-------------------------------------------------"
    write(*, '(A)') "Problem 8: Quartic (x^4) Potential"
```

```fortran
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 0, 4
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob08_driver
```

- **Output:**

```
------------------------------------------------
Problem 8: Quartic (x^4) Potential
------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
------------------------------------------------
The five lowest eigen energies:

n =  0    E = 0.5462783220
n =  1    E = 2.363608793
n =  2    E = 4.670613558
n =  3    E = 7.314949346
n =  4    E = 10.22674160
```

3.9 Problem 9: $V(x) = x^2 + x^4$

- **Potential:** $V(x) = x^2 + x^4$.
- **Quantization Condition Used:** Two Soft Turning Points.

$$\int_{x_L}^{x_R} \sqrt{2m(E - (x^2 + x^4))}\, dx = \left(n + \frac{1}{2}\right)\pi\hbar \quad (n = 0, 1, \cdots)$$

- **Reason:** The potential is continuous and confining on both sides ($x \to \pm\infty$). There are no infinite walls, so the standard WKB rule for two soft turning points applies.
- **Turning Point Calculation:** The potential is given by:

$$V(x) = x^2 + x^4$$

The classical turning points x_{tp} occur where the total energy equals the potential energy:

$$E = V(x_{tp}) \implies x^4 + x^2 - E = 0$$

This is a quadratic equation in terms of x^2. We can solve for x^2 using the quadratic formula:

$$x^2 = \frac{-1 \pm \sqrt{1 - 4(1)(-E)}}{2} = \frac{-1 \pm \sqrt{1 + 4E}}{2}$$

Since x^2 must be non-negative (real coordinate space), we discard the negative root. Thus:

$$x^2 = \frac{\sqrt{1 + 4E} - 1}{2}$$

Taking the square root, we obtain the symmetric turning points $\pm x_R$:

$$\boxed{x_R = \sqrt{\frac{\sqrt{1 + 4E} - 1}{2}}}$$

The integration is performed between $x_L = -x_R$ and x_R.

- **Code:**

```fortran
!==========================================================
! File     : prob09_quadratic_quartic.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Quadratic-quartic potential
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!==========================================================


module prob09_physics
    use constants_mod
    implicit none
    real(dp), parameter :: alpha = 0.5_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = x**2 + x**4
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = -sqrt((-0.5_dp + 0.5_dp * sqrt(1.0_dp + 4*E)))
        x_R = sqrt((-0.5_dp + 0.5_dp * sqrt(1.0_dp + 4*E)))
    end subroutine turning_points

end module prob09_physics


program prob09_driver
    use wkb_toolbox
    use prob09_physics
```

```fortran
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "--------------------------------------------------"
    write(*, '(A)') "Problem 9: Quadratic-quartic potential"
    write(*, '(A)') "--------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "--------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 0, 4
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob09_driver
```

- **Output:**

```
--------------------------------------------------
Problem 9: Quadratic-quartic potential
--------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
--------------------------------------------------
The five lowest eigen energies:

n =  0    E = 0.8433843666
n =  1    E = 3.022273281
n =  2    E = 5.613439059
n =  3    E = 8.505683902
n =  4    E = 11.64263731
```

3.10 Problem 10: One-dimensional Half-Cubic Potential

- **Potential:** $V(x) = x^3$ for $x \geq 0$, ∞ otherwise.
- **Quantization Condition Used:** One Rigid Wall, One Soft Turning Point.

$$\int_0^{x_R} \sqrt{2m(E - x^3)}\, dx = \left(n - \frac{1}{4}\right) \pi\hbar \quad (n = 1, 2, \cdots)$$

- **Reason:** The potential has an infinite barrier (hard wall) at $x = 0$ and a standard soft turning point at x_R determined by the cubic term.

- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = x^3$$

The right turning point $\boxed{x_R = \sqrt[3]{E}}$. Since this is a Half-Cubic potential, the left turning point $\boxed{x_L = 0}$.

- **Code:**

```
!============================================================
! File     : prob10_half_cubic.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Half-cubic potential
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!============================================================


module prob10_physics
    use constants_mod
    implicit none
    real(dp), parameter :: alpha = -0.25_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = x**3
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = 0.0_dp
        x_R = E ** (1.0_dp/3.0_dp)
    end subroutine turning_points

end module prob10_physics


program prob010_driver
    use wkb_toolbox
    use prob10_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "------------------------------------------------
    "
```

```fortran
    write(*, '(A)') "Problem 10: Half-cubic Potential"
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "------------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)') " "

    do n = 1, 5
        energy = find_eigenenergy(n, potential, alpha, turning_points)
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob010_driver
```

- **Output:**

```
------------------------------------------------
Problem 10: Half-cubic Potential
------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
------------------------------------------------
The five lowest eigen energies:

n =  1    E = 2.270340814
n =  2    E = 6.275714865
n =  3    E = 10.79485384
n =  4    E = 15.66228272
n =  5    E = 20.79935503
```

3.11 Problem 11: One-dimensional Half-Quartic Potential

- **Potential:** $V(x) = x^4$ for $x \geq 0$, ∞ otherwise.
- **Quantization Condition Used:** One Rigid Wall, One Soft Turning Point.

$$\int_0^{x_R} \sqrt{2m(E - x^4)}\, dx = \left(n - \frac{1}{4}\right)\pi\hbar \quad (n = 1, 2, \cdots)$$

- **Reason:** The potential has an infinite barrier (hard wall) at $x = 0$ and a standard soft turning point at x_R determined by the quartic term.
- **Turning Point Calculation:** At the turning points, kinetic energy E is equal to potential energy $V(x)$.

$$E = V(x)$$

$$\implies \quad E = x^4$$

The right turning point $\boxed{x_R = \sqrt[4]{E}}$. Since this is a Half-Cubic potential, the left turning point $\boxed{x_L = 0}$.

- **Code:**

```
!==============================================================
! File     : prob11_half_quartic.f90
! Author   : Mohak Das
! Course   : Computational Lab: Quantum Mechanics
! Topic    : WKB approximation for Half-quartic potential
! Units    : m = h_bar = 1
! Method   : WKB quantization + trapezoidal integration
!==============================================================


module prob11_physics
    use constants_mod
    implicit none
    real(dp), parameter :: alpha = -0.25_dp

contains

    function potential(x) result(res)
        real(dp), intent(in) :: x
        real(dp) :: res
        res = x**4
    end function potential

    subroutine turning_points(E, x_L, x_R)
        real(dp), intent(in)  :: E
        real(dp), intent(out) :: x_L, x_R

        x_L = 0.0_dp
        x_R = E ** 0.25
    end subroutine turning_points

end module prob11_physics


program prob11
    use wkb_toolbox
    use prob11_physics
    implicit none

    integer :: n
    real(dp) :: energy

    write(*, '(A)') "--------------------------------------------"
    write(*, '(A)') "Problem 11: Half Quartic potential"
    write(*, '(A)') "--------------------------------------------"
    write(*, '(A, G0.2, A, G0.2)') "Unit used: m = ", mass, ",  h_bar
    = ", h_bar
    write(*, '(A)') "Method: WKB + Trapezoidal rule"
    write(*, '(A)') "--------------------------------------------"
    write(*, '(A)') "The five lowest eigen energies: "; write(*, '(A)
    ') " "
```

```
    do n = 1, 5
        energy = find_eigenenergy(n, potential, alpha, turning_points
    )
        write(*, '(A, I2, A, G0.10)') "n = ", n, "    E = ", energy
    end do

end program prob11
```

- **Output:**

```
------------------------------------------------
Problem 11: Half Quartic potential
------------------------------------------------
Unit used: m = 1.0,  h_bar = 1.0
Method: WKB + Trapezoidal rule
------------------------------------------------
The five lowest eigen energies:

n =  1    E = 2.363578292
n =  2    E = 7.314854584
n =  3    E = 13.36385915
n =  4    E = 20.20830944
n =  5    E = 27.69574842
```

4 Remarks

1. WKB is exact for harmonic oscillator and infinite square well.
2. For linear turning points (triangular well), results differ slightly from exact Airy-function values due to lack of explicit Airy matching.
3. Accuracy improves for higher quantum numbers, as expected.